

# SVYHLEDAT (VLOOKUP) - Tahák - JakNaExcel

<https://office.lasakovi.com/>

Zdrojová tabulka					
	A	B	C	D	E
1	Id	Věk	Jméno	Id2	Přibližné
2	11	22	Eva	11	0 až < 22
3	22	35	Iva	22	22 až < 33
4	33	22	EVA	55	33 až < 44
5	44	27	Jan	33	44 až < 55
6	55	42	Ivan	44	55 až ...

## Syntaxe česky

=SVYHLEDAT (hledat; tabulka; sloupec; [typ])

## Syntaxe anglicky

=VLOOKUP (hledat; tabulka; sloupec; typ)

**hledat** - co hledám - v oblasti první sloupec

**tabulka** - oblast kde hledám, prohledává první sloupec

**sloupec** - číslo sloupce z tabulky s odpovídající hodnotou

**typ** - *volitelně* - NEPRAVDA - přesná, PRAVDA - nejbližší (seřazeno)

Typ	Výsledek	Funkce	Poznámka
Klasika přesná shoda	EVA	=SVYHLEDAT (33; A2: C6; 3; NEPRAVDA)	najde jméno
	22	=SVYHLEDAT (33; A2: C6; 2; NEPRAVDA)	jiný sloupec Věk
	EVA	=SVYHLEDAT (33; \$A\$2: \$C\$6; 3; NEPRAVDA)	zamykat tabulku
	EVA	=SVYHLEDAT (33; A2: C6; 3; 0)	lze použít i nulu
Přibližná shoda	22 až < 33	=SVYHLEDAT (31; A2: E6; 5; PRAVDA)	najde přesnou existuje-li nebo nejbližší nižší
	33 až < 44	=SVYHLEDAT (33; A2: E6; 5; PRAVDA)	
Zástupný znak	11	=SVYHLEDAT ("E*"; C2: D6; 2; NEPRAVDA)	* posloupnost znaků
	11	=SVYHLEDAT ("*a*"; C2: D6; 2; NEPRAVDA)	obsahuje ačko
	22	=SVYHLEDAT ("I*"; C2: D6; 2; NEPRAVDA)	najde první výskyt
	11	=SVYHLEDAT (L24 & "*" ; C2: D6; 2; NEPRAVDA)	v L24 je písmeno E
	11	=SVYHLEDAT ("E??"; C2: D6; 2; NEPRAVDA)	? Právě jedenn znak
Přehození sloupců s využitím funkce ZVOLIT	EVA	{=SVYHLEDAT (33; ZVOLIT ({1\2}; A2: A6; C2: C6) ; 2; NEPRAVDA) }	Zadávat maticově <b>Ctrl + Shift + Enter</b>
	EVA	=SVYHLEDAT (33; ZVOLIT ({1\2}; \$A\$2: \$A\$6; \$C\$2: \$C\$6) ; 2; NEPRAVDA)	v <b>nové verzi Excelu</b> netřeba jako matice
	Jan	{=SVYHLEDAT (33; ZVOLIT ({1\2}; \$D\$2: \$D\$6; \$C\$2: \$C\$6) ; 2; NEPRAVDA) }	prohledávaný sloupec není první
	Jan	{=SVYHLEDAT (33; ZVOLIT ({2\1}; \$C\$2: \$C\$6; \$D\$2: \$D\$6) ; 2; NEPRAVDA) }	přehodit sloupce
	Jan	=XLOOKUP (33; D2: D6; C2: C6)	náhrada SVYHLEDAT
Poslední hodnota	55	=SVYHLEDAT (9E+307; A2: C6; 1; PRAVDA)	poslední hodnota ve sloupci
Přesná shoda Eva vs EVA	11	=SVYHLEDAT ("Eva"; C2: D6; 2; NEPRAVDA)	nerozlišuje velikost písmen
	22	{=SVYHLEDAT (MAX (STEJNÉ ("EVA"; C2: C6) * (A2: A6) ) ; A2: C6; 2; NEPRAVDA) }	záleží na velikost písmen, <b>maticově</b>
Chyby	#NENÍ_K_DISPOZICI	=SVYHLEDAT ("33"; A2: C6; 3; NEPRAVDA)	v čísle hledá text
	#NENÍ_K_DISPOZICI	=SVYHLEDAT (31; A2: C6; 3; NEPRAVDA)	ID 31 neexistuje
	#NENÍ_K_DISPOZICI	=SVYHLEDAT ("33"; A2: C6; 4; NEPRAVDA)	sloupec 4 chybí v tabulce
	#NENÍ_K_DISPOZICI	=SVYHLEDAT ("Eva "; C2: D6; 2; NEPRAVDA)	jméno Eva s mezerou
	#NENÍ_K_DISPOZICI	=SVYHLEDAT (33, 44; C2: D6; 2; NEPRAVDA)	špatně hledat dvě hodnoty
	Iva	=SVYHLEDAT (32; A2: C6; 3)	chyba přibližná shoda
	EVA	=SVYHLEDAT (33; A2: C6; 3)	chyba přibližná shoda
	Eva	=SVYHLEDAT (11; A2: C6; 3)	chyba přibližná shoda
Ošetření chyb	EVA	=SVYHLEDAT (HODNOTA ("33"); A2: C6; 3; 0)	ošetření: Text na číslo
	nenalezeno	=IFERROR (SVYHLEDAT ("33"; A2: C6; 3; NEPRAVDA) ; "nenalezeno")	nezobrazí chybu, ale náš text
Kontrola	33	=SVYHLEDAT (33; \$A\$2: \$C\$6; 1; NEPRAVDA)	vracím sloupec ve kterém hledám
Ve spojení s POZVYHLEDAT	EVA	=SVYHLEDAT (33; A2: C6; POZVYHLEDAT ("Jméno"; A1: C1; 0) ; NEPRAVDA)	funkce POZVYHLEDAT najde číslo sloupce
Více hodnot maticově	22	{=SVYHLEDAT ({11; 22}; A2: D6; 2; NEPRAVDA) }	označit více buněk a vložit maticově
	35		
Více kritérií		A2 & "-" & B2	více kritérií